// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * The default value of {decimals} is 18. To select a different value for
     * {decimals} you should overload it.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless this function is
     * overridden;
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
        }
        _balances[to] += amount;

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        _balances[account] += amount;
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
        }
        _totalSupply -= amount;

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(
        address owner,
        address spender,
        uint256 amount
    ) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/interfaces/IBasePriceOracle.sol



// SPDX-License-Identifier: Apache-2.0
pragma solidity 0.8.17;

interface IBasePriceOracle {
    event AssetStatusSet(address indexed baseAsset, bool indexed isEnabled);

    function supportsAsset(address, address) external view returns (bool);

    function getPrice(address, address) external view returns (bool, uint256);

    function setAssetStatus(address, bool) external;
}

// File: contracts/interfaces/IOracleManager.sol



// SPDX-License-Identifier: Apache-2.0
pragma solidity 0.8.17;

import "./IBasePriceOracle.sol";

interface IOracleManager {
    event SetOracles(address indexed _emitter, IBasePriceOracle[] _oracles);

    function setOracles(IBasePriceOracle[] memory) external;

    function getPrice(address) external view returns (uint256);

    function getOracles() external view returns (IBasePriceOracle[] memory);
}

// File: contracts/interfaces/IPool.sol



// SPDX-License-Identifier: Apache-2.0
pragma solidity 0.8.17;

interface IPool {
    struct Account {
        uint256 notional;
    }

    event Initialized(
        address indexed _router,
        address indexed _borrower,
        address[] _collateralAssets,
        address indexed _lentAsset
    );

    event OracleSet(address indexed _emitter, address _oracle);

    function leftoversWithdrawn() external view returns (bool);

    function startsAt() external view returns (uint32);

    function activeAt() external view returns (uint32);

    function maturesAt() external view returns (uint32);

    function coupon() external view returns (uint96);

    function ltv() external view returns (uint96);

    function originationFee() external view returns (uint96);

    function minSupply() external view returns (uint256);

    function maxSupply() external view returns (uint256);

    function supply() external view returns (uint256);

    function borrowed() external view returns (uint256);

    function borrower() external view returns (address);

    function whitelistedLender() external view returns (address);

    function lentAsset() external view returns (address);

    function collateralAssets(uint256) external view returns (address);

    function collateralReserves(address) external view returns (uint256);

    function notionals(address) external view returns (uint256);

    function initialize(
        address _borrower,
        address _lentAsset,
        address[] memory _collateralAssets,
        uint96 _coupon,
        uint96 _ltv,
        uint96 _originationFee,
        uint32 _activeAt,
        uint32 _maturesAt,
        uint256 _minSupply,
        uint256 _maxSupply,
        address _whitelistedLender
    ) external;

    function deposit(address, uint256) external;

    function supplyCollateral(address, uint256) external;

    function borrow(address, uint256) external;

    function repay(uint256) external;

    function redeem(address) external;

    function _default(address) external;

    function withdrawLeftovers(uint256) external;

    function getCollateralAssets() external view returns (address[] memory);
}

// File: contracts/interfaces/IPoolFactory.sol



// SPDX-License-Identifier: Apache-2.0
pragma solidity 0.8.17;

interface IPoolFactory {
    event PoolCreated(
        uint256 indexed _pid,
        address[] _collateralAssets,
        address indexed _lentAsset,
        uint256 _upfrontCoupon
    );

    event ParametersChanged(
        uint256 indexed _maxNumberOfCollateralAssets,
        uint96 indexed _originationFee
    );

    function createPool(
        address _lentAsset,
        address[] memory _collateralAssets,
        uint96 _coupon,
        uint96 _ltv,
        uint32 _activeAt,
        uint32 _maturesAt,
        uint256 _minSupply,
        uint256 _maxSupply,
        address _whitelistedLender
    ) external returns (address pool);

    function setMaxNumberOfCollateralAssets(
        uint256 _maxNumberOfCollateralAssets
    ) external;

    function setOriginationFee(uint96 _originationFee) external;

    function router() external view returns (address);

    function pid() external view returns (uint256);

    function maxNumberOfCollateralAssets() external view returns (uint256);

    function originationFee() external view returns (uint96);

    function MAX_ORIGINATION_FEE() external view returns (uint96);

    function pidToPoolAddress(uint256) external view returns (address);

    function getAllPools() external view returns (address[] memory);
}

// File: contracts/interfaces/IRouter.sol



// SPDX-License-Identifier: Apache-2.0
pragma solidity 0.8.17;

import "./IPoolFactory.sol";
import "./IOracleManager.sol";

interface IRouter {
    event FactorySet(address indexed _emitter, address indexed _poolFactory);

    event OracleManagerSet(
        address indexed _emitter,
        address indexed _oracleManager
    );

    event TreasurySet(address indexed _emitter, address indexed _treasury);

    event Deposit(
        uint256 indexed _pid,
        address indexed _asset,
        uint256 indexed _amt
    );

    event Borrow(
        uint256 indexed _pid,
        uint256 indexed _borrAmt,
        address[] collateralAssets,
        uint256[] _colAmts
    );

    event Repay(
        uint256 indexed _pid,
        address indexed _lentAsset,
        uint256 indexed _amt
    );

    event Redeem(
        uint256 indexed _pid,
        address indexed _asset,
        bool indexed _hasDefaulted
    );

    event LeftoversWithdrawn(uint256 indexed _pid, uint256 indexed _amt);

    function poolFactory() external view returns (IPoolFactory);

    function oracleManager() external view returns (IOracleManager);

    function treasury() external view returns (address);

    function getBorrowingPower(
        address[] calldata _collateralAssets,
        uint256[] calldata _amts
    ) external view returns (uint256 _borrowingPower);

    function deposit(uint256 _pid, uint256 _amt) external;

    function borrow(
        uint256 _pid,
        address[] calldata _collateralAssets,
        uint256[] calldata _amts
    ) external;

    function repay(uint256 _pid, uint256 _amt) external;

    function redeem(uint256 _pid) external;

    function withdrawLeftovers(uint256 _pid) external;
}

// File: contracts/Router.sol



// SPDX-License-Identifier: Apache-2.0
pragma solidity 0.8.17;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/Pausable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "./interfaces/IBasePriceOracle.sol";
import "./interfaces/IOracleManager.sol";
import "./interfaces/IPoolFactory.sol";
import "./interfaces/IPool.sol";
import "./interfaces/IRouter.sol";

/**
 * @title Router
 * @author LombardFi
 * @notice The router is the entry point for interacting with the LombardFi system.
 * @dev Much of the verification logic is done in the router to compress the size of the Pool contract.
 */
contract Router is IRouter, Pausable, Ownable, ReentrancyGuard {
    using SafeERC20 for IERC20;

    /**
     * @notice Address of the PoolFactory.
     */
    IPoolFactory public poolFactory;

    /**
     * @notice Address of the OracleManager.
     */
    IOracleManager public oracleManager;

    /**
     * @notice Address of the protocol treasury.
     * @dev Receives the origination fee at pool creation.
     */
    address public treasury;

    /**
     * @notice Verify that an integer is greater than 0.
     * @dev Throws an error if the uint256 is equal to 0
     * @param amt The integer to check.
     */
    modifier nonZero(uint256 amt) {
        require(amt > 0, "Router::zero amt");
        _;
    }

    /**
     * @notice Verify that an address is not the zero address.
     * @dev Throws an error if the address is the zero address.
     * @param _address The address to check.
     */
    modifier nonZeroAddress(address _address) {
        require(_address != address(0), "Router::zero address");
        _;
    }

    /**
     * @notice Set the PoolFactory implementation address. Callable only by the owner.
     * @dev Throws an error if the supplied address is the zero address.
     * @param _poolFactory The new implementation.
     */
    function setFactory(IPoolFactory _poolFactory)
        external
        nonZeroAddress(address(_poolFactory))
        onlyOwner
    {
        require(
            address(poolFactory) == address(0),
            "Router::factory already set"
        );
        poolFactory = _poolFactory;
        emit FactorySet(msg.sender, address(_poolFactory));
    }

    /**
     * @notice Set the OracleManager implementation address. Callable only by the owner.
     * @dev Throws an error if the address is the zero address.
     * @param _oracleManager The new implementation.
     */
    function setOracleManager(IOracleManager _oracleManager)
        external
        nonZeroAddress(address(_oracleManager))
        onlyOwner
    {
        oracleManager = _oracleManager;
        emit OracleManagerSet(msg.sender, address(_oracleManager));
    }

    /**
     * @notice Set the treasury address. Callable only by the owner.
     * @dev Throws an error if the address is the zero address.
     * @param _treasury The new recipient.
     */
    function setTreasury(address _treasury)
        external
        nonZeroAddress(_treasury)
        onlyOwner
    {
        treasury = _treasury;
        emit TreasurySet(msg.sender, _treasury);
    }

    /**
     * @notice Pause the contract. Callable only by the owner.
     */
    function pause() external onlyOwner {
        _pause();
    }

    /**
     * @notice Unpause the contract. Callable only by the owner.
     * @dev The contract must be paused to unpause it.
     */
    function unpause() external onlyOwner {
        _unpause();
    }

    /**
     * @notice Deposit into a pool.
     * @dev The contract must not be paused.
     * `_amt` must be nonzero.
     * A pool with the `_pid` must exist.
     * Caller must not be the pool's borrower.
     * Caller must be the whitelisted lender if the pool has one.
     * Caller must have approved this contract to spend `_amt` of the pool's lent asset.
     * The pool must not be active or mature.
     * The pool must have sufficient open capacity for `_amt`.
     * @param _pid The id of the pool to deposit in.
     * @param _amt The amount of the pool's lent asset to deposit.
     */
    function deposit(uint256 _pid, uint256 _amt)
        external
        nonZero(_amt)
        whenNotPaused
        nonReentrant
    {
        IPool pool = _getPool(_pid);

        // If there is a whitelisted lender verify that the sender is the whitelisted lender.
        address whitelistedLender = pool.whitelistedLender();
        require(
            whitelistedLender == address(0) || whitelistedLender == msg.sender,
            "Router::caller not whitelisted lender"
        );
        // Caller cannot be borrower
        _verifyCallerIsNotBorrower(pool);
        // Pool must accept deposits
        require(
            block.timestamp < pool.activeAt(),
            "Router::expired deposit period"
        );
        // Pool must have enough open capacity
        require(
            pool.maxSupply() - pool.supply() >= _amt,
            "Router::supply cap exceeded"
        );

        // Transfer lent asset from lender to pool and perform accounting.
        address lentAsset = pool.lentAsset();
        IERC20(lentAsset).safeTransferFrom(msg.sender, address(pool), _amt);
        pool.deposit(msg.sender, _amt);

        emit Deposit(_pid, lentAsset, _amt);
    }

    /**
     * @notice Borrow the avaiable amount of lent asset from a pool.
     * Transfers the pool's lent asset from pool to borrower.
     * Transfers collateral from borrower to pool.
     * @dev The contract must not be paused.
     * A pool with the `_pid` must exist.
     * Caller must be the pool's borrower.
     * Lengths of `_collateralAssets` and `amts` must match.
     * `_collateralAssets` must be the pool's collateral assets or a subset.
     * The pool must be active.
     * The pool's minimum deposit must have been reached.
     * The loan amount and value of the supplied collateral must satisfy the loan-to-value ratio.
     * @param _pid The id of the pool to borrow from.
     * @param _collateralAssets The assets to deposit as collateral
     * @param _amts The amounts corresponding to the collaterals
     */
    function borrow(
        uint256 _pid,
        address[] calldata _collateralAssets,
        uint256[] calldata _amts
    ) external whenNotPaused nonReentrant {
        IPool pool = _getPool(_pid);

        // Only the borrower can borrow
        require(msg.sender == pool.borrower(), "Router::caller not borrower");
        // Cardinality of assets and amounts must match
        require(
            _collateralAssets.length == _amts.length,
            "Router::invalid params"
        );
        // Assets must be a subset of the collateral assets
        require(
            _assetsAreValidPoolCollateral(pool, _collateralAssets),
            "Router::invalid assets"
        );

        // Pool must be active and the minimum must be met
        uint256 supply = pool.supply();
        require(
            block.timestamp >= pool.activeAt() &&
                block.timestamp < pool.maturesAt() &&
                supply >= pool.minSupply(),
            "Router::can't borrow"
        );

        uint256 borrowAmount = supply - pool.borrowed();

        // Calculate the loan value from the oracle
        address lentAsset = pool.lentAsset();
        uint256 loanValue = (oracleManager.getPrice(lentAsset) * borrowAmount) /
            10**ERC20(lentAsset).decimals();

        // Calculate the value of the collateral from the oracle
        uint256 borrowingPower = getBorrowingPower(_collateralAssets, _amts);

        // Check that the loan-to-value ratio is not exceeded
        require(
            (loanValue * 10**18) / borrowingPower <= pool.ltv(),
            "Router::low BP"
        );

        // Transfer collateral from borrower and perform accounting for the deposits
        for (uint256 i = 0; i < _collateralAssets.length; ) {
            if (_amts[i] > 0) {
                IERC20(_collateralAssets[i]).safeTransferFrom(
                    msg.sender,
                    address(pool),
                    _amts[i]
                );
                pool.supplyCollateral(_collateralAssets[i], _amts[i]);
            }
            unchecked {
                ++i;
            }
        }

        // Transfer the loan from pool to borrower and perform accounting for the borrow
        pool.borrow(lentAsset, borrowAmount);

        emit Borrow(_pid, borrowAmount, _collateralAssets, _amts);
    }

    /**
     * @notice Repay a part of the loan.
     * Transfers the pool's lent asset from borrower to pool.
     * Transfers collateral from pool to borrower.
     * @dev A pool with the `_pid` must exist.
     * `_amt` must be nonzero.
     * The contract must not be paused.
     * Pool must be active if the router is under normal operation.
     * Caller must be the pool's borrower.
     * The pool must be active.
     * @param _pid The id of the pool to repay in.
     * @param _amt The amount of the pool's lent asset to repay.
     */
    function repay(uint256 _pid, uint256 _amt)
        external
        nonZero(_amt)
        nonReentrant
    {
        IPool pool = _getPool(_pid);

        // Pool must be active.
        // If the Router is paused then repayment is allowed at any time.
        require(
            paused() ||
                (block.timestamp >= pool.activeAt() &&
                    block.timestamp < pool.maturesAt()),
            "Router::not active"
        );

        // Check that there is existing debt
        uint256 debt = pool.borrowed();
        require(debt > 0, "Router::no debt");

        // The amount repaid is the minimum of the supplied `_amt` and the outstanding debt
        // The owner cannot repay more than the debt (which will lead to accounting errors and overflows)
        // This nullifies griefing attacks (since anyone can repay on behalf of the borrower)
        uint256 amountToRepay;
        if (_amt < debt) {
            amountToRepay = _amt;
        } else {
            amountToRepay = debt;
        }

        // Perform accounting and transfer collateral from the pool to the borrower
        pool.repay(amountToRepay);

        // Transfer lent asset from the borrower to the pool
        address lentAsset = pool.lentAsset();
        IERC20(lentAsset).safeTransferFrom(
            msg.sender,
            address(pool),
            amountToRepay
        );

        emit Repay(_pid, lentAsset, _amt);
    }

    /**
     * @notice Redeem notional and yield from a mature pool or redeem notional from an unsuccessful pool.
     * Transfers the pool's lent asset from pool to caller.
     * Transfers collateral from caller to pool.
     * @dev The contract must not be paused.
     * A pool with the `_pid` must exist.
     * Caller must not be the pool's borrower.
     * The pool must be mature or active with less deposits than the minimum.
     * Caller must have made a deposit.
     * Caller can redeem only once per pool.
     * @param _pid The id of the pool to redeem from.
     */
    function redeem(uint256 _pid) external whenNotPaused nonReentrant {
        IPool pool = _getPool(_pid);

        // Caller cannot be the borrower
        _verifyCallerIsNotBorrower(pool);

        // Either the pool is mature (redeem notional + yield)
        // Or the pool is active but minimum has not been met (redeem notional)
        require(
            block.timestamp >= pool.maturesAt() ||
                (block.timestamp >= pool.activeAt() &&
                    pool.minSupply() > pool.supply()),
            "Router::can't redeem"
        );

        // Caller has deposited something and has not redeemed their rewards yet
        uint256 notional = pool.notionals(msg.sender);
        require(notional > 0, "Router::no notional");

        bool hasDefaulted = pool.borrowed() > 0;

        if (!hasDefaulted) {
            // All debt has been repaid
            // Perform accounting and transfer lent asset from pool to lender
            pool.redeem(msg.sender);
        } else {
            // There is outstanding debt (borrower defaults)
            // Perform accounting and transfer pro-rata lent asset and collateral from pool to lender
            pool._default(msg.sender);
        }

        emit Redeem(_pid, pool.lentAsset(), hasDefaulted);
    }

    /**
     * @notice Withdraw redundant yield from a pool.
     * Transfers a part of the upfront for the unrealized size back to the borrower.
     * @dev The contract must not be paused.
     * A pool with the `_pid` must exist.
     * Caller must be the pool's borrower.
     * Borower can withdraw only once.
     * The pool must be active.
     * @param _pid The id of the pool to withdraw leftovers from.
     */
    function withdrawLeftovers(uint256 _pid)
        external
        whenNotPaused
        nonReentrant
    {
        IPool pool = _getPool(_pid);

        // Caller must be the pool's borrower
        require(pool.borrower() == msg.sender, "Router::not borrower");

        // Borower can withdraw only once per pool
        require(!pool.leftoversWithdrawn(), "Router::already withdrawn");

        // Pool must be active
        require(block.timestamp >= pool.activeAt(), "Router::not active");

        // If the minimum was not met, withdraw all of the upfront rewards
        uint256 supplyClaim = pool.maxSupply();
        uint256 _supply = pool.supply();
        if (_supply >= pool.minSupply()) {
            // If the minimum was met, withdraw the rewards for the unfilled size
            supplyClaim -= _supply;
        }

        // Perform accounting and transfer the rewards back to the borrower
        uint256 rewardsToWithdraw = (pool.coupon() * supplyClaim) / 10**18;
        pool.withdrawLeftovers(rewardsToWithdraw);

        emit LeftoversWithdrawn(_pid, rewardsToWithdraw);
    }

    /**
     * @notice Utility function that returns the value of collateral.
     * Prices are fetched from the OracleManager.
     * @dev Also used for off-chain data retrieval.
     * @param _collateralAssets Array of collateral assets.
     * @param _amts The amounts corresponding to the collateral assets.
     * @return _borrowingPower The total value of the collateral.
     */
    function getBorrowingPower(
        address[] calldata _collateralAssets,
        uint256[] calldata _amts
    ) public view returns (uint256 _borrowingPower) {
        uint256 numCollaterals = _collateralAssets.length;
        for (uint256 i = 0; i < numCollaterals; ) {
            // Skip asset if the amount is 0
            if (_amts[i] > 0) {
                // Fetch the price of a unit of asset from the oracle
                address asset = _collateralAssets[i];
                uint256 price = oracleManager.getPrice(asset);

                // Calculate the value of the collateral and add it to the borrowing power.
                _borrowingPower +=
                    (price * _amts[i]) /
                    10**ERC20(asset).decimals();
            }
            unchecked {
                ++i;
            }
        }
    }

    /**
     * @notice Utility function that checks whether an array of addresses match pool collateral.
     * They must be a subset.
     * @param _pool The pool to check the assets against.
     * @param _assets The array of ERC20 token addresses to check against the pool.
     * @return Whether the given assets are valid subset of pool collateral.
     */
    function _assetsAreValidPoolCollateral(
        IPool _pool,
        address[] memory _assets
    ) private view returns (bool) {
        // Reads all assets from storage into memory.
        address[] memory collateralAssets = _pool.getCollateralAssets();
        uint256 numCollaterals = collateralAssets.length;

        // Iterate through the assets
        uint256 assetsLength = _assets.length;
        for (uint256 i = 0; i < assetsLength; ) {
            address asset = _assets[i];

            // Check if there is a duplicate asset in the rest of the array
            // Only need to search after the current position
            uint256 j = i + 1;
            for (; j < assetsLength; ) {
                if (asset == _assets[j]) {
                    return false;
                }
                unchecked {
                    ++j;
                }
            }

            // Check if the collateral exists in the pool
            for (j = 0; j < numCollaterals; ) {
                if (asset == collateralAssets[j]) {
                    break;
                }
                unchecked {
                    ++j;
                }
            }

            if (j == numCollaterals) {
                // Did not find a match
                return false;
            }

            unchecked {
                ++i;
            }
        }
        return true;
    }

    /**
     * @notice Private function that verifies that the caller is not the pool's borrower.
     * @param _pool The pool contract.
     */
    function _verifyCallerIsNotBorrower(IPool _pool) private view {
        require(msg.sender != _pool.borrower(), "Router::cannot be borrower");
    }

    /**
     * @notice Private function that gets a pool address from a pool id.
     * @dev Throws an error if a pool with the `_pid` does not exist.
     * @param _pid The id of the pool to get the address of.
     * @return _pool The pool contract.
     */
    function _getPool(uint256 _pid) private view returns (IPool _pool) {
        address poolAddress = poolFactory.pidToPoolAddress(_pid);
        require(poolAddress != address(0), "Router::no pool");
        _pool = IPool(poolAddress);
    }
}